KERNFORSCHUNGSZENTRUM KARLSRUHE

Juni 1970 Externer Bericht 3/70-6

Institut für Experimentelle Kernphysik

FORTRAN-Program for the computation of the Surface Impedance of Superconductors

J. Halbritter

Gesellschaft für Kernforschung, m.b.H., Karlsruhe

GESELLSCHAFT FÜR KERNFORSCHUNG M. B. H.
KARLSRUHE

KERNFORSCHUNGSZENTRUM KARLSRUHE

Juni 1970 Externer Bericht 3/70-6

Institut für Experimentelle Kernphysik

FORTRAN-Program for the computation of the Surface Impedance of Superconductors

J. Halbritter

Gesellschaft für Kernforschung, m.b.H., Karlsruhe

Abstract

The FORTRAN-program presented in this report computes the surface impedance Z of superconductors in the Meissner state for frequencies lower than the energy gap. The formulae used are exact within the frame work of the BCS-theory for weak-coupling superconductors. Strong coupling effects can be introduced by using measured Δ/kT_c values. In the program the temperature dependence of the gap parameter $\Delta(T)$ is approximated by $(\Delta(T)/\Delta(0))^2 = \cos \frac{\pi}{2} (T/T_c)^2$.

I. SUMMARY

The program (see tables) computes the surface impedance Z for the actual parameters of measurements:

[FO] : frequency; [TE(..)] : temperature,

the symbols in the brackets [...] are used in the FORTRAN-program (see tables). The number [IT] of temperature values is limited to $\leq$ 20 by the program.

The material parameters for a superconductor in the weak coupling limit are

[TC] : critical temperature;
[FAK] : observed $\Delta(0)/kT_c$ value;
[DLON] : London penetration depth at zero temperature and infinite mean free path ℓ:

$$\delta_L = \sqrt{m/(\mu_o e^2 N)}$$

with m the mass, e the charge and N the density of (transport) electrons;

[XKOH] : dimension of a cooper pair (coherence length) at T = 0 and $\ell = \infty$,

$$\xi_F = \hbar\, v_F \,/\, 2\Delta(0) \qquad (2)$$

with the Fermi velocity v_F;

[FREI] : mean free path of the electrons in the normal state.

In the program the unit for temperature is degree Kelvin and the unit for length Angstrom.

The accuracy of computation can be increased by lowering the factor [REST]. The lower limit of this factor in our program (see tables) is given by about $5 \cdot 10^{-7}$ because the number of abcissas [IQ] is limited by 10^3. For REST $\sim 10^{-4}$ the absolut accuracy of Z will be about 1 %, the relative accuracy (for normal materials like lead or niobium) will be about 1 ‰. For a further increase of accuracy one should use adapted abcissas.

With the described input data the program computes the surface impedance Z = R + iX for specular "S" and diffuse "D" reflection of the charge carriers at the surface. In the output the surface resistance R [RS] or [RD] is printed in Ohm, whereas the surface reactance $X = \omega\mu_o\lambda$ is given through the penetration depth $\lambda = \int_0^\infty \mathrm{Re}\{H(z,t)/H(0,t)\}dz$ [XS] or [XD].

As an example the following input-data:

```
2
                                                              (card 1)

 3.E   +9   4.17    2.
                                                              (card 2)

7.2        2.05       280.       1750.       17500.
                                                              (card 3)

1.      E-3
                                                              (card 4)
```



result in:

```
.3000E 100.4170E 010.2000E 010.1000E-02
 SURFACE IMPEDANCE IN THE SUPERCONDUCTING STATE
========================================================

 MATERIALPARAMETERS:

 TC = 7.20 K         GAP(T=0)/KTC = 2.05
LONDON PEN. DEPTH(T=0,1/L=0) =   280.0 ANGSTROM
COHERENCE LENGTH(T=0,1/L=0)F =  1750.0 ANGSTROM
MEAN FREE PATH                = 17500.0 ANGSTROM

 FOR F = 0.3000E 10 HERTZ THE BCS-THEORY RESULTS

               SPECULAR-            DIFFUSE REFLEXION

   TEM-        PEN.    SURFACE     PEN.   SURFACE
 PERATURE     DEPTH   RESISTANCE   DEPTH  RESISTANCE
   (K)         (A)      (OHM)       (A)     (OHM)

   4.17       379.0   0.430E-05    437.4  0.432E-05          0.1354E 01

   2.00       354.5   0.108E-06    410.3  0.111E-06          0.1266E 01
```

The program (see tables) can be used also for a computation of the penetration for $\omega = 0$ (or $\hbar\omega << \Delta$) if one omits the subroutine NONAN in the program-deck.

II. INTRODUCTION

In plane geometry the surface impedance $Z = R + iX$ is given by the penetration of the magnetic field $H(z,t)$ into the wall:

$$Z = i\omega\mu_o \int_o^\infty dz\, H(z,t)/H(0,t) = i\omega\mu_o\, A(0,t)/B(0,t) \qquad (3)$$

with z the distance from the surface and A the vector potential: $\mu_o \vec{H} = \nabla\times\vec{A}$. The fields $H(z,t)$ or $A(z,t)$ are the solutions of the following equations and boundary conditions [1,3-5].

Maxwell's equation ($\partial^2/\partial t^2$ can be neglected for frequencies lower than 10^{13} Hz):

$$-\partial^2 A(z,t)/\partial t^2 = \mu_o\, j(z,t);$$

material equation:

$$j(z,t) = -\int_{-\infty}^{o} dt' \int dz' Q(z-z',\, t+t')\, A(z',t');$$

boundary conditions for the fields:

$$\partial A(z,t)/\partial z\,|_{z=0} = B(0,t),\quad A(\infty,t) = 0;$$

boundary conditions for the charge carriers at the surface:

specular or diffuse reflection.

According to [3] with the Fourier transformed integral kernel $Q(k,\omega)$ the surface impedance can then be written:

$$\text{specular reflection:}\quad Z_s = i\mu_o\omega \frac{1}{\pi}\int_{-\infty}^{\infty} \frac{dk}{k^2+\mu_o Q(k,\omega)}$$

$$\text{diffuse reflection:}\quad Z_D = i\mu_o\omega \frac{\pi}{\int_0^{\infty} dk\,\ell n\{1+\mu_o Q(k,\omega)/k^2\}} \qquad (4)$$

For Q evaluation of (4) the k-integration is transformed in a q-integration by: $q = k\ \xi_F\ \Delta(0)/\Delta(T)$ [DX(...)], i.e. the momenta k will be measured in units of a coherence length. With the abreviations

$$\delta_L(T,\ell,\omega) = 1/\sqrt{\mu_o|Q(0,\omega)|} \qquad [ENP]$$

$$\kappa = \delta_L(T,\ell,\omega)\ /\{\xi_F\Delta(0)/\Delta(T)\} \qquad [AK]$$

(4) can be written:

$$Z_s = i\mu_o\omega\ \delta_L(T,\ell,\omega) \frac{2\kappa}{\pi}\int_0^{\infty} \frac{dq}{\kappa^2 q^2+Q(q,\omega)/|Q(0,\omega)|}$$

$$Z_D = i\mu_o\omega \frac{\pi\ \delta_L(T,\ell,\omega)}{\kappa\int_0^{\infty} dq\,\ell n(1+Q(q,\omega)/(|Q(0,\omega)|\kappa^2 q^2))} \qquad (5)$$

Before we can start with the integration dq in subroutine HAUPT and compute the integrand in subroutine INTQ (...) the integralkernel $Q(k,\omega)$ should be evaluated. According to [1] we split Q in an in ω analytical term Q_A and a non-analytical term Q_D:

$$\mu_o \delta_L^2 \, Q(k,\omega) = \bar{Q}_A(k,\omega) + \bar{Q}_D(k,\omega) \tag{6}$$

The analytical term Q_A is defined by:

$$\bar{Q}_A(k,\omega) = a \sum_{\substack{m=(2n+1)a \\ n \geq 1}} \left\{ \frac{1-m(m+i\tilde{\omega})+\sqrt{m^2+1}\sqrt{(m+i\tilde{\omega})^2+1}}{\sqrt{m^2+1}\,\sqrt{(m+i\tilde{\omega})^2+1}} \right. \tag{7}$$

$$\left. 4\, J\left[\frac{\sqrt{m^2+1}+\sqrt{(m+i\tilde{\omega})^2+1}}{2} + \frac{\hbar v_F}{2\Delta(T)\ell} \right] + c.c. \right\}$$

($a = \pi kT/\Delta(T)$; $\tilde{\omega} = \hbar\omega/\Delta(T)$: [0]). $\bar{Q}_A$ is computed in subroutine ISUM, $\bar{Q}_A$ is purely real and therefore gives for $\tilde{\omega} << 1$ the penetration depth, i.e. $\bar{Q}_A$ describes the current of "Cooper" pairs. $\bar{Q}_D$ which disappears for $\tilde{\omega} \to 0$ and $T \to 0$ consists of a real term $\bar{Q}_P$ and a term $\bar{Q}_{SCH}$, describing the rf absorption:

$$\bar{Q}_D(k,\omega) = \bar{Q}_P(k,\omega) + \bar{Q}_{SCH}(k,\omega) \tag{8}$$

$$\bar{Q}_P(k,\omega) = \int_{1-\tilde{\omega}}^{1} d\bar{\omega} \left\{ \mathrm{th}\frac{\Delta(T)}{2k_BT}\,\bar{\omega} - \mathrm{th}\frac{\Delta(T)}{2k_BT}\,(\bar{\omega}+\tilde{\omega}) \right\} \cdot \tag{9}$$

$$\left\{ \frac{1+\bar{\omega}(\bar{\omega}-\tilde{\omega})+i\sqrt{-(\bar{\omega}-\tilde{\omega})^2+1}\sqrt{\bar{\omega}^2-1}}{\sqrt{\bar{\omega}^2-1}\,\sqrt{-(\bar{\omega}-\tilde{\omega})^2+1}}\, J\left[\frac{\sqrt{-(\bar{\omega}-\tilde{\omega})^2+1}+i\sqrt{\bar{\omega}^2-1}}{2} + \frac{\hbar v_F}{2\Delta(T)\cdot\ell} \right] + c.c. \right\}$$

$$\bar{Q}_{SCH}(k,\omega) = 2i \int_1^\infty d\bar{\omega} \left\{ \text{th} \frac{\Delta(T)}{2k_BT} \bar{\omega} - \text{th} \frac{\Delta(T)}{2k_BT} (\bar{\omega} + \tilde{\omega}) \right\} \cdot$$

$$\cdot \left\{ \frac{1+\bar{\omega}(\bar{\omega}+\tilde{\omega})+\sqrt{\bar{\omega}^2-1}\sqrt{(\bar{\omega}+\tilde{\omega})^2-1}}{\sqrt{\bar{\omega}^2-1}\ \sqrt{(\bar{\omega}+\tilde{\omega})^2-1}} J \left[-i \frac{\sqrt{(\bar{\omega}+\tilde{\omega})^2-1}+\sqrt{\bar{\omega}^2-1}}{2} + \frac{\hbar v_F}{2\Delta(T)\ell} \right] \right. \qquad (10)$$

$$\left. \frac{1+\bar{\omega}(\bar{\omega}+\tilde{\omega})-\sqrt{\bar{\omega}^2-1}\sqrt{(\bar{\omega}+\tilde{\omega})^2-1}}{\sqrt{\bar{\omega}^2-1}\ \sqrt{(\bar{\omega}+\tilde{\omega})^2-1}} \text{Re } J \left[(-i \frac{\sqrt{(\bar{\omega}+\tilde{\omega})^2-1}+\sqrt{\bar{\omega}^2-1}}{2} + \frac{\hbar v_F}{2\Delta(T)\ell} \right] \right\}$$

$\bar{Q}_P$ and $\bar{Q}_{SCH}$ will be computed in subroutine NONAN.

The integrals (7), (9) and (10) depend on the function J which results from an angle integration:

$$J(u) = 6 \frac{1}{\hbar v_F k} \left\{ -\frac{u\ 2\Delta(T)}{\hbar v_F k} + \left[\frac{u2\Delta(T)}{\hbar v_F k}^2 + 1 \right] \text{arctg} \frac{\hbar v_F k}{u2\Delta(T)} \right\} \qquad (11)$$

J(u) will be evaluated in subroutine WINK (...).

In the following sections we will discuss in more detail the various summations or integrations in the subroutines.

III. MOMENTUM INTEGRATION

The q-integration is done by Simpson's rule (formula 25.4.6 in [6]). The q-range $\{0,\infty\}$ is subdivided into the intervals [I] = 1, ..., 5 {[G(1)], [G(2)]} ..4} and the 6th inter-

val {[G(6)] 1/[G(1)]}, in which q is substituted by $\frac{1}{q}$. The abcissas [DX ( )] within these intervals are evaluated in the main program, their number is given by [ID(I)] (with [ID(I)] = u and [DELT] = h in 25.4.6 in [6]. The subdivision into 6 intervals before using Simpson's rule has been introduced to improve the accuracy by giving certain q-regions larger weight. For materials with, e.g. $\delta_L/\xi_F \ll 1$ small momenta $k \approx (\delta_L \xi_F)^{-1/2}$ are most important. After evaluation of Q in subroutines NONAN and ISUM the integrand is calculated in subroutine INTQ (,) and summed in subroutine HAUPT. The result - $Z/(\omega\mu_o\delta_L(T,\ell,\omega))$ - is transfered to the main program, where after some multiplications the surface resistance in Ohm and the penetration depth $\lambda(T,\ell,\omega)$ in Angstrom are printed out.

IV. EVALUATION OF $Q(k,\omega)$

We start with the smallest fraction of Q, namely the $\bar{Q}_D$ in subroutine NONAN to reduce rounding errors, The larger $\bar{Q}_A$ is computed afterwards in subroutine ISUM. A numerical computation of $\bar{Q}_D$ with Simpson's rule is not very precise because of the square root singularity of the integrand. More adequate for numerical integration of square root sigularities is the integration by Chebyshev Polynomials of the First Kind, i.e. the use of formula 25.4.39 of [6]:

$$\int_a^b \frac{f(y)dy}{\sqrt{(y-a)(b-y)}} = \sum_{i=1}^{n} \omega_i f(y) + R_n \qquad (12)$$

$$y_i = \frac{b+a}{2} + \frac{b-a}{2} x_i, \quad x_i = \cos\frac{(2i-1)}{2n}, \quad \omega_i = \frac{\pi}{n}$$

The evaluation of the residue R_n for (9) and (10) is too difficult and it is omitted, therefore. A sufficient n is chosen according to the requirement that there must be enough abcissas (> 5) in the region where f(y) has a sharp maximum.

IV.1 $\bar{Q}_{SCH}(k,\omega)$

This term describes the rf absorption, i.e. $\bar{Q}_{SCH}$ includes the imaginary part of Q. For computation integrand of (10) is transformed by substituting $u = 1/\bar{\omega}$; i.e. the interval (a,b) in (12) is: (0,1). For the integral the maximum of $f(y) \propto 1/\sqrt{1+u\tilde{\omega}-u}$ becomes more important for lower $\tilde{\omega}$, i.e. the number of abcissas [IS] will increase with $1/\sqrt{\tilde{\omega}}$.

IV.2 $\bar{Q}_P(k,\omega)$

For $\tilde{\omega} < 1$ $\bar{Q}_P(k,\omega)$ is a slowly varying function with $\bar{\omega}$ the number of abcissas [IP] will therefore be small. Moreover there is no need of a more exact computation of $\bar{Q}_P$ for $\tilde{\omega} \ll 1$, because $\bar{Q}_P$ is real and small compared to $\bar{Q}_A$.

IV.3 $\bar{Q}_A(k,\omega)$

$\bar{Q}_A$ (7) is computed in subroutine ISUM. For $\tilde{\omega} = 0$ and $T = 0$ ($\tilde{\omega} \leq 2$) Q_A is the only non-vanishing part. For the numerical computation the temperature sum in ISUM yields easily a good accuracy. The number of terms in the sum [MS] must be increased for $T \to 0$ because the terms are $\propto 1/(T/T_c)^2$. To reduce rounding errors in computing the sum ISUM starts with the smallest terms.

REFERENCES

1. J. Halbritter
Externer Bericht 3/69-2, (Kernforschungszentrum Karlsruhe , 1969)

2. T.P. Sheahen
Phys. Rev. 149, 368 (1966)

3. G.E.H. Reuter, E.H. Sondheimer
Proc. Roy. Soc. A 195, 336 (1948)
R.B. Dingle
Physica XIX, 311 (1953)

4. I.M. Khalatnikov, A.A. Abrikosov
Adv. in Phys. 8, 45 (1958)

5. J.P. Turneaure
Thesis (HEPL-report-507, Stanford, 1967)

6. M. Abramowitz, I.E. Stegun
Handbook of Mathematical Functions (U.S. Government Printing Office, Washington, C.D., 1964)

FORTRAN - PROGRAM

```
C       SURFACE IMPEDANCE OF SUPERCONDUCTORS FOR FREQUENCIES SMALLER THAN
C       THE ENERGY GAP.
        DIMENSION G(6),ID(6),TE(20)
        REAL*8 DELT,DET,DQ(6),DA,DE,DZ,DX(1000),PI        ,DQ1 ,DS(1000)
        COMPLEX*8 CHS,CHD,CI
        COMPLEX*16 CDS(1000),CT(1000),CS,CD
        COMMON/HL1/ID,IQ,MS,IS,IP
        COMMON/HL2/AK,A2,FL,O,GP
        COMMON/HL3/PI,DQ,DX,DQ1,DS
        COMMON/HL4/CHS,CHD,CI
        COMMON/HL5/CDS,CT,CS,CD

C       IT=NUMBER OF TEMPERATUREVALUES.
        READ (5,1) IT
C       FO=FREQUENCY (HERTZ),TE(I)=TEMPERATURES (KELVIN).
        READ (5,2) FO,(TE(I),I=1,IT)
C       MATERIALPARAMETERS(UNITS:DEGREE KELVIN,ANGSTROM)
        READ (5,2) TC,FAK,DLON,XKOH,FREI
C       REST GIVES THE ACCURACY OF THE COMPUTATION.
        READ (5,2) REST
    1 FORMAT(I5)
    2 FORMAT(8E10.4)
        WRITE(6,2) FO,(TE(I),I=1,IT),REST
C       THE MOMENTUMINTEGRAL IS SPLITTED IN 6 PARTS OVER THE INTERVALLS (
C       (I),G(I+1)).
        G(1)=1.E-6
        G(2)=1.E-2
        G(3)=0.2
        G(4)=2.
        G(5)=5.
        G(6)=50.
C       THESE 6 INTEGRALS ARE EVALUATED BY SIMPSON'S RULE. 1/ID(I) GIVES
C       THE SPACING OF THE ABCISSAS.
        ID(1)=5
        ID(2)=7
        ID(3)=15
        ID(4)=20
        ID(5)=20
        ID(6)=9
        IS=(6.E-4/REST)**0.25+1.
        DO 38 N=1,6
   38 ID(N)=ID(N)*IS
        PI=3.1415926535
        PH =0.479*10.**(-10)
        CI=(0.,1.)
        FLO=XKOH/FREI
        AL=DLON/XKOH
        OG=PH*FO/TC/FAK
        DX(1)=0.
        DX(2)=G(1)
        IQ=2
        DO 40 N=1,5
        LL=ID(N)-1
        D=ID(N)*2
        ID(N)=LL
        DA=G(N)
        DZ=G(N+1)
        DELT=(DZ-DA)/D
```

```
      DET=DELT*2.
      DQ(N)=DELT/3.
      DO 39 K=1,LL
      DA=DA+DET
      DE=DA-DELT
      IQ=IQ+1
      DX(IQ)=DE
      IQ=IQ+1
   39 DX(IQ)=DA
      DE=DZ-DELT
      IQ=IQ+1
      DX(IQ)=DE
      IQ=IQ+1
   40 DX(IQ)=DZ
      LL=ID(6)-1
      D=ID(6)*2
      ID(6)=LL
      DA=G(1)
      DZ=1./G(6)
      DELT=(DZ-DA)/D
      DET=DELT*2.
      DQ(6)=DELT/3.
      IQ=IQ+1
      DX(IQ)=1./DA
      DO 41 K=1,LL
      DA=DA+DET
      DE=DA-DELT
      IQ=IQ+1
      DX(IQ)=1./DE
      IQ=IQ+1
   41 DX(IQ)=1./DA
      DE=DZ-DELT
      IQ=IQ+1
      DX(IQ)=1./DE
      IQ=IQ+1
      DX(IQ)=1./DZ
      WRITE(6,100)
      WRITE(6,101)
      WRITE(6,102)
      WRITE(6,103)TC,FAK,DLON,XKOH,FREI
      WRITE(6,107)FO
      WRITE(6,108)
      DO 22 K=1,IT
      T=TE(K)/TC
      B=PI/2.*(1.-T*T)
      AG=SQRT(SIN(B))
      FL=FLO/AG
      O=OG/AG
      IF(O-0.5)69,22,22
   69 GP=AG/T*FAK
      IS=1./SQRT(O*REST)/5.
      IP=1./SQRT((2.-O)*REST)/5.
      DO 19 I=1,IQ
      DS(I)=0.
   19 CDS(I)=0.
      CALL NONAN
      MS=GP/(SQRT(REST)*2.*PI)
      CALL ISUM
```

```
      DQ1=CDABS(CDS(1))
      ENP=1./DSQRT(DQ1)
      AK=AL*ENP*AG
      A2=AK*AK
      CALL HAUPT
      CHS=CHS*ENP
      CHD=CHD*ENP
      XS=REAL(CHS)*DLON
      XD=REAL(CHD)*DLON
      B=(FO/10.**17)*8.*PI*PI*DLON
      RS=AIMAG(CHS)*B
      RD=AIMAG(CHD)*B
      WRITE(6,109) TE(K),XS,RS,XD,RD,CHS,CHD
   22 CONTINUE
  100 FORMAT(1H1,48H SURFACE IMPEDANCE IN THE SUPERCONDUCTING STATE )
  101 FORMAT(1H0,48H================================================)
  102 FORMAT(1H0/2X,19HMATERIALPARAMETERS:)
  103 FORMAT(1H0,6H TC = ,F4.2,24H K        GAP(T=0)/KTC = ,F4.2/32H LOND
     1DON PEN. DEPTH(T=0,1/L=0) = ,F7.1,9H ANGSTROM/32H COHERENCE LENGTH
     2(T=0,1/L=0)F = ,F7.1,9H ANGSTROM/32H MEAN FREE PATH
     3= ,F7.1,9H ANGSTROM)
  107 FORMAT(1H0,9H FOR F = ,E10.4,29H HERTZ THE BCS-THEORY RESULTS/3H
     1 /14X,9HSPECULAR-,10X,17HDIFFUSE REFLEXION)
  108 FORMAT(1H0,3X,4HTEM-,6X,4HPEN.,4X,7HSURFACE,5X,4HPEN.,4X,7HSURFACE
     1/2X,8HPERATURE,3X,5HDEPTH,3X,10HRESISTANCE,2X,5HDEPTH,3X,10HRESIST
     2ANCE/4X,3H(K),7X,3H(A),6X,5H(OHM),6X,3H(A),6X,5H(OHM))
  109 FORMAT(1H0,3X,F4.2,3X,F8.1,1X,E10.3,1X,F8.1,1X,E10.3,8X,4E11.4)
      STOP
      END
```

```
      SUBROUTINE HAUPT
      DIMENSION ID(6)
      REAL*8 DQ(6),DX(1000),PI,        DQ1,DS(1000)
      COMPLEX*8 CHS,CHD,CI
      COMPLEX*16 CDS(1000),CS            ,CT(1000),CIS,CID,CES,CED,CAS,
     1CAD,CD
      COMMON/HL1/ID,IQ,MS,IS,IP
      COMMON/HL2/AK,A2,FL,O,GP
      COMMON/HL3/PI,DQ,DX,DQ1,DS
      COMMON/HL4/CHS,CHD,CI
      COMMON/HL5/CDS,CT,CS,CD
      M=1
      I=2
      CALL INTQ(I,M)
      CIS=DX(2)*CS
      CID=DX(2)*CD+2.*DX(2)
      DO 40 J=1,6
      IF(J-6)11,10,10
   10 M=2
      CALL INTQ(I,M)
      CIS=CIS+CS*DX(2)
      CID=CID+CD*DX(2)
   11 CES=0.
      CED=0.
      CAS=0.5*CS
      CAD=0.5*CD
      LL=ID(J)
      DO 39 K=1,LL
      CALL INTQ(I,M)
      CES=CES+CS
      CED=CED+CD
      CALL INTQ(I,M)
      CAS=CAS+CS
      CAD=CAD+CD
   39 CONTINUE
      CALL INTQ(I,M)
      CES=CES+CS
      CED=CED+CD
      CALL INTQ(I,M)
      CIS=CIS+DQ(J)*(4.*CES+(2.*CAS+CS))
   40 CID=CID+DQ(J)*(4.*CED+(2.*CAD+CD))
      A=2./PI*AK
      CHS=A*CIS
      CHD=2./A/CID
      RETURN
      END
```

```
      SUBROUTINE INTQ(L,M)
      DIMENSION ID(6)
      REAL*8 DQ(6),DX(1000),PI,        DX2,DA,    DQ1,DS(1000)
      COMPLEX*8 CHS,CHD,CI
      COMPLEX*16 CDS(1000),CT(1000),CS,CD,CDG
      COMMON/HL1/ID,IQ,MS,IS,IP
      COMMON/HL2/AK,A2,FL,O,GP
      COMMON/HL3/PI,DQ,DX,DQ1,DS
      COMMON/HL4/CHS,CHD,CI
      COMMON/HL5/CDS,CT,CS,CD
      CDG=CDS(L)/DQ1
      DX2=DX(L)*DX(L)
      DA=DX2*A2
      GO TO (10,11),M
   10 CS=1./(DA+CDG)
      CD=CDLOG(1.+CDG/DA)
      GO TO 9
   11 CS=1./(A2+CDG/DX2)
      CD=CDLOG(1.+CDG/DA)*DX2
    9 L=L+1
      RETURN
      END
```

```
      SUBROUTINE NONAN
      DIMENSION ID(6)
      REAL*8 D ,DX(1000),PI,        DQ1,DS(1000),DV,DZ,DU,DG,DA,DC,DP,DM,
     1DW,DWO,DQ(6), OT,O2
      COMPLEX*8 CHS,CHD,CI
      COMPLEX*16 CDS(1000),CT(1000),CS,CD,CY
      COMMON/HL1/ID,IQ,MS,IS,IP
      COMMON/HL2/AK,A2,FL,O,GP
      COMMON/HL3/PI,DQ,DX,DQ1,DS
      COMMON/HL4/CHS,CHD,CI
      COMMON/HL5/CDS,CT,CS,CD
      OT=O/2.*GP
      IF(CT-0.2)10,10,9
   10 O2=CT*OT
      DV=OT*(1.+O2/6.*(1.+O2/20.*(1.+O2/42.*(1.+O2/72.*(1.+O2/110.)))))*
     1DEXP(-OT)
      CT=-CT*2.
      GO TO 8
    9 OT=-CT*2.
      DV=(1.-DEXP(OT))/2.
    8 B=4*IS
      BS=PI/B
      DO 48 K=1,IS
      B=4*(IS-K)+2
      D=B*BS
      DZ=0.5*(1.+DCOS(D))
      DU=GP/DZ
      DA=DEXP(-DU)
      DG=DEXP(-DU+OT)
      DC=DA/((1.+DG)*(1.+DA))*DV
      DP=DZ+1.
      DM=1.-DZ
      DU=DP*DM
      DW=DSQRT(DU)
      DA=DZ*O
      DWO=DSQRT(DU+DA*(2.+DA))
      DP=DZ*DSQRT(DZ*DP)
      DU=DA/DU
      IF(DU-0.1)7,7,6
    7 DU=DU*(2.+DA)/2.
      DM=DU*(1.-0.5*DU*(1.-DU*(1.-1.25*DU*(1.-1.4*DU*(1.-1.5*DU*(1.-11.*
     1DU/7.*(1.-1.5*DU*(1.-13.*DU/9*(1.-1.4*DU))))))))))*DW
      GO TO 5
    6 DM=DWO-DW
    5 CY=-CI*DM/2./DZ+FL
      CALL WINK(CY)
      DU=DC/DP/DWO
      DWU=DWO*DW
      DM=1.+DZ*(DZ+O)
      DPS=(DM+DWU)*DU
      DMS=(DM-DWU)*DU
      DO 20 I=1,IQ
   20 CDS(I)=CDS(I)+DPS*CT(I)
      DM=DWO+DW
      CY=CI*DM/2./DZ+FL
      CALL WINK(CY)
```

```
      DO 21 I=1,IQ
   21 DS(I)=DS(I)+DMS*CT(I)
   48 CONTINUE
      CY=-CI*BS*4.
      DO 22 I=1,IQ
      CDS(I)=(CDS(I)+DS(I))*CY
   22 DS(I)=0.
      B=4*IP
      BS=PI/B
      DO 47 K=1,IP
      B=4*(IP-K)+2
      D=B*BS
      DZ=1.+0.5*O*(1.+DCOS(D))
      DU=GP*DZ
      DA=DEXP(-DU)
      DG=DEXP(-DU-OT)
      DC=DG/((1.+DG)*(1.+DA))*DV
      DP=DZ+1.
      DM=DZ-1.
      DW=DSQRT(DP*DM)
      DA=O-DM
      DU=DP-O
      DWO=DSQRT(DA*DU)
      CY=DWO+CI*DW+FL
      CALL WINK(CY)
      CY=(1.+DZ*(DZ-O)+CI*DW*DWO)/DSQRT(DP*DU)*DC
      DO 23 I=1,IQ
   23 DS(I)=DS(I)+CY*CT(I)
   47 CONTINUE
      B=+BS*4.
      DO 24 I=1,IQ
   24 CDS(I)=CDS(I)+DS(I)*B
      RETURN
      END
```

```
      SUBROUTINE ISUM
      DIMENSION ID(6)
      REAL*8 DQ(6),DX(1000),PI, G,     DU,DW,DS(1000),DQ1
      COMPLEX*8 CHS,CHD,CI
      COMPLEX*16 CDS(1000),CS            ,CB,CU,CW,CL,CP,CA,CT(1000),CD
      COMMON/HL1/ID,IQ,MS,IS,IP
      COMMON/HL2/AK,A2,FL,O,GP
      COMMON/HL3/PI,DQ,DX,DQ1,DS
      COMMON/HL4/CHS,CHD,CI
      COMMON/HL5/CDS,CT,CS,CD
      G=PI/GP
      DO 20 I=1,IQ
   20 DS(I)=0.
      DO 30 M=1,MS
      B=2*(MS-M)+1
      B=B*G
      DU=B*B+1.
      CB=B+O*CI
      CU=CB*CB+1.
      DW=DSQRT(DU)
      CW=CDSQRT(CU)
      CL=(DW+CW)/2.+FL
      CP=DW*CW
      CA=(1.-B*CB+CP)/CP
      CALL WINK(CL)
      DO 21 I=1,IQ
   21 DS(I)=DS(I)+CA*CT(I)
   30 CONTINUE
      DO 22 I=1,IQ
   22 CDS(I)=DS(I)*G +CDS(I)
      RETURN
      END
```

```
      SUBROUTINE WINK(CL)
      DIMENSION ID(6)
      REAL*8 DQ(6),DX(1000),PI,        D,DQ1,DS(1000)
      COMPLEX*8 CHS,CHD,CI
      COMPLEX*16 CDS(1000),CS            ,CT(1000),CD,CC,C2,CL
      COMMON/HL1/ID,IQ,MS,IS,IP
      COMMON/HL2/AK,A2,FL,O,GP
      COMMON/HL3/PI,DQ,DX,DQ1,DS
      COMMON/HL4/CHS,CHD,CI
      COMMON/HL5/CDS,CT,CS,CC
      DO 20 I=1,IQ
      D=DX(I)
      CD=D/CL
      C2=CD*CD
      U=CDABS(CD)
      IF(U-0.2)10,10,11
   10 CT(I)=3./CL*(1./3.-C2*(1./15.-C2*(1./35.-C2*(1./63.-C2*(1./99.-
     1C2*(1./143.-C2*(1./195.-C2/255.))))))
      GO TO 20
   11 CD=1./CD
      CT(I)=0.75/D*(-2.*CD+(1.+CD*CD)/CI*(CDLOG(CD+CI)-CDLOG(CD-CI)))
   20 CONTINUE
      RETURN
      END
```